

## ¡Vamos! Protección social

### Tablero de indicadores del Convenio entre la República del Paraguay y la Unión Europea (Ley 6650/2020)

Instrucciones para el cálculo del Indicador #3 "Porcentaje de población de 18 a 29 años que no está cursando ni está graduada en estudios de nivel terciario o universitario que ha concluido cursos de capacitación laboral"

1. Acceder al portal de microdatos del INE, en su sección de EPHC serie comparable, base de población:  
<https://www.ine.gov.py/datos/encuestas/eph/Poblacion/>
2. Por defecto, ingresará en la última base disponible, pero se sugiere confirmar en "seleccionar año" que se opta por la base correspondiente a EPHC del año que desee obtener la estimación.
3. En la caja "descarga" (a la derecha) seleccionar "SAV" para descargar la base en formato de SPSS (compatible con los comandos enviados).

\*En base de microdatos de población (REG02) de la Encuesta Permanente de Hogares Continua, Serie comparable.

\*Indicador 3 .

\*Condición de graduación de Educación superior

```
RECODE ED06C (15=1) (1 thru 13=1) INTO ed06REC.  
VARIABLE LABELS ed06REC 'Condición de tenencia de título educación superior'.  
EXECUTE.
```

```
RECODE ed06REC (SYSMIS=0).  
EXECUTE.
```

```
VALUE LABELS ed06REC  
0 'Sin título de educación superior'  
1 'Con título de educación superior'.
```

\*Condición de estudiante de educación superior.

```
IF (ED08 >= 12 & ED08 <= 18) ESTUDIANTESUPERIOR=1.  
EXECUTE.
```

```
RECODE ESTUDIANTESUPERIOR (SYSMIS=0).  
EXECUTE.
```

```
VALUE LABELS ESTUDIANTESUPERIOR  
0 'No estudiante de educación superior'  
1 'Estudiante de educación superior'.
```

\*Condición de graduado o estudiante de educación superior.

```
IF (ed06REC = 0 & ESTUDIANTESUPERIOR = 0) EDUCNOSUPER=0.  
EXECUTE.
```

```
IF (ed06REC = 1 OR ESTUDIANTESUPERIOR = 1) EDUCNOSUPER=1.  
EXECUTE.
```

VARIABLE LABELS EDUCNOSUPER 'Condición de graduado o estudiante de educación superior'.

```
VALUE LABELS EDUCNOSUPER  
0 'No Graduado/a ni estudiante de educación superior'  
1 'Graduado/a o estudiante de educación superior'.
```

\*Condición de cursante finalizado de cursos de educación no formal (excepto idioma).

```
IF (ED12=1 & (ED13>=2 & ED13<=11) & (ED14A=1)) CURSOS=1.  
EXECUTE.
```

```
IF (SYSMIS(CURSOS)) CURSOS=0.  
EXECUTE.
```

VARIABLE LABELS CURSOS 'Condición de cursante finalizado de cursos de educación no formal (excepto idioma)'.

```
VALUE LABELS CURSOS  
0 'Sin cursos no formales finalizados'  
1 'Con cursos no formales finalizados'.
```

\*Ponderación de base de datos.  
WEIGHT BY FEX.

\*Solicitud de frecuencia sobre universo definido en el indicador.

```
USE ALL.  
COMPUTE filter_$=(P02 >= 18 AND P02 <= 29 AND EDUCNOSUPER=0).  
VARIABLE LABELS filter_$ 'P02 >= 18 AND P02 <= 29 AND EDUCNOSUPER=1 (FILTER)'.  
VALUE LABELS filter_$ 0 'Not Selected' 1 'Selected'.  
FORMATS filter_$ (f1.0).  
FILTER BY filter_$.  
EXECUTE.
```

```
FREQUENCIES VARIABLES=CURSOS  
/ORDER=ANALYSIS.
```

```
FILTER OFF.  
USE ALL.  
EXECUTE.
```